

# GeoShop Verrechnungsmodul

## Zusammenfassung

Diese Dokumentation beschreibt den GeoShop Verrechnungsmodul.

Die Dokumentation darf nur mit Erlaubnis der infoGrips GmbH vervielfältigt werden.

# Inhaltsverzeichnis

1. Einleitung .....	4
1.1. Aufbau des Handbuchs .....	4
1.2. Ergänzende Dokumentationen .....	4
1.3. Konventionen .....	4
2. Installation .....	5
2.1. Überblick .....	5
2.2. Programmdateien installieren .....	5
2.3. Lizenzinformationen eingeben .....	5
3. Funktionsweise .....	6
3.1. Überblick .....	6
3.2. Datenfluss .....	6
3.3. Skriptparameter .....	7
3.4. Skriptresultat .....	8
4. Einrichten von Tarifzonen .....	9
4.1. Überblick .....	9
4.2. Konfiguration von Tarifzonen .....	9
4.3. Konfiguration der Punktebene .....	10
5. Aufbau eines Preisskript .....	12
5.1. Überblick .....	12
5.2. Skriptparameter .....	12
5.3. iG/Script Bibliothek price.lib .....	12
5.4. Beispiele .....	13
5.5. Angabe des Preisskript in der Produktdatei .....	15
A. Literaturverzeichnis .....	16

# 1. Einleitung

Der **GeoShop Verrechnungsmodul** ist ein Erweiterungsmodul zum Produkt GeoShop. Mit dem GeoShop Verrechnungsmodul können Verrechnungsfunktionen (= Tarife) für Geodatenbestellungen frei definiert werden. Der Verrechnungsmodul zeichnet sich durch folgende Eigenschaften aus:

- Definition von beliebigen Verrechnungsfunktionen (= Tarife).
- Online Anzeige des Preis im GeoShop Client Applet.
- Für jedes Datenprodukt kann eine eigene Verrechnungsfunktion definiert werden.
- Online Verschnitt der Bestellfläche mit einem Tarifzonennetz (z.B. Flächenanteil pro Zone) bzw. einer Punktebene (z.B. Anzahl Fixpunkte vom Typ LFP3).
- Es können einfache (z.B. nach Fläche und Anzahl Ebenen) oder komplexe Tarife definiert werden (z.B. Abrechnung nach Modell Buschor).
- etc.

## 1.1. Aufbau des Handbuchs

Dieses Handbuch ist wie folgt aufgebaut:

- In Kapitel 2 wird die Installation des GeoShop Verrechnungsmoduls beschrieben.
- In Kapitel 3 wird die Funktionsweise des Verrechnungsmoduls beschrieben.
- In Kapitel 4 ist wird die Einrichtung von Tarifzonen beschrieben.
- In Kaptitel 5 ist der Aufbau eines Preisskripts beschrieben.
- Im Anhang ist das Literaturverzeichnis enthalten.

## 1.2. Ergänzende Dokumentationen

Das Benutzerhandbuch wird durch folgende Dokumentationen ergänzt:

- GeoShop Benutzerhandbuch [1].
- GeoEditor Benutzerhandbuch [2].
- iG/Script Benutzerhandbuch und Referenzhandbuch [3].

## 1.3. Konventionen

In dieser Dokumentation werden folgende Konventionen eingehalten:

<i>Kursiv</i>	Namen von Dateien und URL's
<b>fett</b>	neue Begriffe, Namen von Funktionen oder Methoden
<code>courier</code>	Programmtext oder Eingaben im Betriebssystem

## 2. Installation

### 2.1. Überblick

Dieses Kapitel beschreibt die Installation des GeoShop Verrechnungsmodul. Es wird vorausgesetzt, dass das Produkt GeoShop bereits korrekt installiert ist (s.a. [1]). Bei neueren GeoShop Versionen wird der Verrechnungsmodul bereits vorinstalliert mitgeliefert, es muss daher lediglich die Lizenzinformation eingetragen werden (s.a. 2.3). Ältere GeoShop Installationen können wie folgt mit dem Verrechnungsmodul nachgerüstet werden:

- Programmdateien installieren.
- Lizenzinformation eingeben.

### 2.2. Programmdateien installieren

Entzippen Sie die mitgelieferte Datei price.zip im Verzeichnis `GEOSHOP_DIR\system`. Bestehende Dateien müssen überschrieben werden.

### 2.3. Lizenzinformationen eingeben


Die Lizenzinformationen müssen in der Datei `\user\license\price.lic` eingetragen werden.

## 3. Funktionsweise

### 3.1. Überblick

Ein Tarif ist im allgemeinen eine beliebige Funktion (Verrechnungsfunktion) welche sich aus den Inputdaten und bzw. den vom Benutzer gewählten Bestellparametern zusammensetzt (z.B. Produkt, Ausschnitt, Ebenen, etc.).

Abbildung 1. Parameterübergabe an Verrechnungsfunktion




Die Verrechnungsfunktion kann einfach sein, oft hängt sie jedoch von vielen unterschiedlichen Parametern ab:

- Den gewählten Inputdaten (z.B. amtl. Vermessung oder Zonenplan).
- Der bestellten Fläche.
- Dem bestellten Datenprodukt (DXF, SHP, MIF/MID, INTERLIS/ITF etc.).
- Dem Benutzer (z.B. normaler Tarif für gelegentlicher Benutzer wie z.B. Architekten bzw. Spezialrabatt für Dauerbezüger wie amtl. Stellen).
- etc.

### 3.2. Datenfluss

Im GeoShop wird die Verrechnungsfunktion durch eine iG/Script Skript (s.a. [3]) implementiert. Mit iG/Script erhält man die notwendige Flexibilität um beliebige Tarife (inkl. MwSt. etc.) berechnen zu können. Der Skript wertet dazu die Parameter aus, welche dem Skript vom GeoShop Client Applet via GeoShop übermittelt werden. Die folgende Graphik zeigt den Datenfluss bei einer online Preisberechnung:

**Abbildung 2. Datenfluss bei der Preisberechnung**

Erläuterungen zur letzten Graphik:

- Der Benutzer selektiert die Daten mit dem GeoShop Client Applet und klickt auf die Funktion Preis berechnen.
- Das Applet schickt alle Bestellparameter (Produkt, Ausschnitt, Ebenen, etc.) an den GeoShop.
- Der GeoShop ruft den für das Produkt definierten Preisskript auf.
- Der Preisskript berechnet auf Grund der Bestellparameter den Preis und gibt diese an den GeoShop zurück.
- Der GeoShop leitet die Preisberechnung weiter an das GeoShop Client Applet (Preisantwort).
- Das GeoShop Client Applet zeigt den Preis an.

**Abbildung 3. Preisantwort im GeopShop Client**

### 3.3. Skriptparameter

Dem Preisskript werden alle Bestellparameter übergeben dazu gehören (abschliessende Aufzählung in 5.2):

- Der Name des Datenprodukts.

- Die Namen der gewählten Ebenen.
- Der Name des aktuellen Benutzers.
- etc.

Zusätzlich können vom GeoShop (vor dem Aufruf des Preisskripts) noch folgende zusätzliche Parameter aus den Daten berechnet werden:

- Ein Verschnitt der Bestellfläche mit einem beliebigen Flächennetz (= Tarifzonen, z.B. Zoneneinteilung). Der Anteil jeder Zonenart wird dem Preisskript als Liste übergeben.
- Ein Verschnitt mit einer beliebigen Punktebene. Dem Preisskript wird eine Liste der Anzahl Punkte pro Punktart übergeben.

Damit die zusätzlichen Parameter berechnet werden, muss die Tarifzone bzw. die Punktebene entsprechend konfiguriert werden (s.a. Kapitel 4).

## 3.4. Skriptresultat

Der Preisskript muss das Resultat in einer Textdatei (`price.txt`) zurück liefern. Die ersten zwei Zeilen sind reserviert und müssen wie folgt gefüllt werden:

- 1. Zeile: Währung, z.B. SFr.
- 2. Zeile: Preis (ohne Währung), z.B. 185.50.

Das GeoShop Client Applet zeigt die Resultatdatei dem Benutzer an (bis auf die ersten beiden Zeilen). Die ersten beiden Zeilen werden zur Bestellung in der Bestelldatenbank (`server/orders`) gespeichert, falls der Benutzer die Bestellung auch tatsächlich auslöst.

## 4. Einrichten von Tarifzonen


### 4.1. Überblick

Der GeoShop kann den Bestellperimeter mit Tarifzonen (Flächennetz) und/oder einer Punktebene verschneiden. Dazu muss die Tarifzone bzw. die Punktebene als Layer mit dem GeoEditor definiert werden. Die Tarifzonen- bzw. Punktdaten wird nach dem Upload der Tarifzonen- bzw. Punktdaten in eine .geo Datei umgewandelt. Die .geo Datei ermöglicht den *schnellen* Online-Verschnitt der Tarifzonen- bzw. Punktdaten mit dem Bestellperimeter.

### 4.2. Konfiguration von Tarifzonen

Tarifzonen werden wie folgt konfiguriert:

1. Öffnen Sie mit dem GeoEditor die Konfiguration `\system\script\il2geo\tarif.cfg` (s.a. [2]) und klicken Sie auf das Objekt `Tarifobjekte.Flaeche_Area`.



2. Erzeugen Sie zum Objekt `Flaechе_Area` mit dem GeoEditor ein `PolygonRate` Objekt. Die einzelnen Felder müssen wie folgt ausgefüllt werden:

**Geometry**

Geometrie der Fläche.

**Layer**

Z.B. Tarif. Der Name des Layer muss dem Preisskript als Parameter übergeben werden (s.a. 5.3).

**Rate Class**

Text nach welchem die einzelnen Flächen gruppiert werden sollen (normalerweise Artcode der Fläche).


## Bemerkungen:

- Der Skript `\system\script\il2geo\tarif.cfg` basiert auf dem Datenmodell `Tarif.ili`. Die entsprechenden Flächen müssen als Datensatz gemäss Datenmodell `Tarif.ili` auf den GeoShop geladen werden.
- Für die Definition von Tarifzonen muss nicht unbedingt das Modell `Tarif.ili` benutzt werden. Falls die Zonen bereits in den Daten vorhanden sind, z.B. in Form einer Gemeindegrenze, kann das `PolygonRate` Objekt auch direkt für das entsprechende Objekt im Datenmodell definiert werden.
- Die Definition von Tarifzonen ist optional. Der Layer muss nur definiert werden, falls der Preis vom Verschnitt des Bestellpolygon mit den Tarifzonen abhängt.

## 4.3. Konfiguration der Punktebene

Punktebenen für die Preisfunktion werden wie folgt konfiguriert:

1. Öffnen Sie mit dem GeoEditor die Konfiguration `\system\script\il2geo\tarif.cfg` (s.a. [2]) und klicken Sie auf das Objekt `Tarifobjekte.Punkt`.



2. Erzeugen Sie zum Objekt `Punkt` mit dem GeoEditor ein `PointRate` Objekt. Die einzelnen Felder müssen wie folgt ausgefüllt werden:

**Geometry**

Punktgeometrie.

**Layer**

Z.B. `Tarif`. Der Name des Layer muss zusammen mit dem Preisfunktion zur Produktdefinition gespeichert werden (.pro Datei, s.a. 5.3).

**Rate Class**

Text nach welchem die einzelnen Punkte gruppiert (d.h. gezählt) werden sollen (normalerweise Artcode des Punkt).

Bemerkungen:

- Der Skript `\system\script\il2geo\tarif.cfg` basiert auf dem Datenmodell `Tarif.ili`. Die entsprechenden Punkte müssen als Datensatz gemäss Datenmodell `Tarif.ili` auf den GeoShop geladen werden.
- Für die Definition der Punktebene muss nicht unbedingt das Modell `Tarif.ili` benutzt werden, falls die Punkte bereits in einem anderen Datenmodell vorhanden sind (z.B. in AV93).
- Die Definition der Punktebene ist optional. Der Layer muss nur definiert werden, falls der Preis von der Anzahl der bestellten Punkte abhängt.

## 5. Aufbau eines Preisskript

### 5.1. Überblick

Der Preisskript erzeugt aus den Bestellparametern die Preisdatei (`price.txt`). Für die Abfrage der Parameter steht eine iG/Script Bibliothek zur Verfügung. Die einzelnen Parameter bzw. die Bibliothek sind im folgenden beschrieben. Am Schluss des Kapitels ist ein vollständiges Beispiel angegeben.

### 5.2. Skriptparameter

Dem Preisskript werden vom GeoShop folgende Parameter in der Map `OPT` übergeben:

Parameter	Beschreibung
<code>product</code>	Produktname.
<code>area</code>	Effektiv bestellte Fläche (d.h. nach dem Schnitt mit den Tarifzonen).
<code>fence_area</code>	Fläche des Bestellpolygon.
<code>user</code>	Benutzername.
<code>prefix</code>	Benutzerprefix (falls vorhanden).
<code>topics</code>	Ausgewählte Datenebenen als kommaseparierte Liste.
<code>polygon_rate</code>	Name des Textfiles, welches die Schnittresultate mit den Tarifzonen enthält.
<code>point_rate</code>	Name des Textfiles, welches die Schnittresultate mit der Punkzebene enthält.
<code>temp_dir</code>	Name des Verzeichnis in welchem temporäre Dateien abgelegt werden können. Wird vom GeoShop nach der Preisberechnung automatisch gelöscht.
<code>log</code>	Name der Logdatei.
<code>output</code>	Name der Textdatei in welchem das Resultat der Berechnung gespeichert werden soll. Die ersten beiden Zeilen haben eine spezielle Bedeutung. 1. Zeile: Währung, 2. Zeile: Preis. Der Rest der Outputdatei wird dem Benutzer im GeoShop Client Applet angezeigt.



Für das Auslesen der Parameter `OPT.polygon_rate` bzw. `OPT.point_rate` stehen spezielle Prozeduren in der Bibliothek `\script\price.lib` zur Verfügung.

### 5.3. iG/Script Bibliothek `price.lib`

Folgende Prozeduren aus der Bibliothek `\script\price.lib` können für die Berechnung des Preis bzw. für das Schreiben der Preisdatei benutzt werden:

`PRICE_DISPLAY_INPUT [][]`

Zeigt alle Inputparameter in der Logdatei an.

`PRICE_CREATE_OUTPUT [][]`

Erstellt eine neue, leere Preisdatei.

**PRICE\_WRITELN [s line][ ]**

Schreibt eine Zeile in die Preisdatei.

**PRICE\_CLOSE\_OUTPUT [ ] [ ]**

Schliesst die Preisdatei.

**PRICE\_SET\_LOG [s logfile][ ]**

Setzt die Logdatei auf einen bestimmten fixen Pfad (vorallem für Debuggingzwecke nützlich).

**PRICE\_GET\_TOPICS [ ] [ ]**

Füllt die vordefinierte Map PRICE\_TOPICS mit Einträgen der Art <Ebene> => TRUE (bzw. DEFAULT => FALSE). Nach dem Aufruf von PRICE\_GET\_TOPICS kann über die Map PRICE\_TOPICS sehr einfach festgestellt werden, ob eine bestimmte Datenebene vom Benutzer ausgewählt wurde oder nicht.

**PRICE\_COUNT\_TOPICS [ ] [i count]**

Gibt die Anzahl ausgewählter Datenebenen zurück.

**PRICE\_GET\_POLYGON\_RATES [ ] [ ]**

Füllt die vordefinierte Map PRICE\_POLYGON\_RATES mit Einträgen für jede Flächenart (Rate Class im GeoEditor). Der Wert für jede Flächenart ist als Faktor zwischen 0.0 und 1.0 zu verstehen (DEFAULT => 0.0). xy => 0.5 bedeutet z.B., dass 50% der Bestellfläche in der Flächenart xy liegen.

**PRICE\_GET\_POINT\_RATES [ ] [ ]**

Füllt die vordefinierte Map PRICE\_POINT\_RATES mit Einträgen für jede Punktart (Rate Class im GeoEditor). Als Wert für jede Punktart wird die Anzahl Punkte in der Bestellfläche eingetragen (DEFAULT => 0).

## 5.4. Beispiele

Das Beispielskript `\user\script\price\price.cfg` berechnet folgende Preisfunktion:

- Für Flächen  $\leq 10$  ha wird eine Grundgebühr von SFr. 20.- und pro Datenebene ein Zuschlag von SFr. 5.- erhoben.
- Für Flächen ] 10.0 .. 300.0] ha wird eine Grundgebühr von SFr. 100.- und pro Datenebene ein Zuschlag von SFr. 25.- erhoben.
- Für Flächen  $> 300$  ha wird eine Grundgebühr von SFr. 200.- und ein pro Datenebene ein Zuschlag von SFr. 50.- erhoben.
- Es wird kein Schnitt mit Tarifzonen oder einer Punktebene gemacht.

Nachfolgend ist der vollständige Quelltext des Beispielskript `price.cfg` angegeben (Prozeduren der Bibliothek `price.lib` sind fett markiert):

```
|LICENSE \license\geoshop.lic
|INCL \script\price.lib

! Hilfsfunktion für die Formatierung von Preisen
PROCEDURE FORMAT_PRICE ! [r price][s result]
  'SFr. ' SWAP TO_STRING 8 RJUST APP
END_PROCEDURE

! Hilfsfunktion für die Berechnung des Tarifs
PROCEDURE CALCULATE_PRICE ! [][r price]
OPT.fence_area TO_INT 0.0001 * TO_INT => OPT.ha
```

```

! der Tarif ist abhängig von der Anzahl ausgewählter Ebenen
PRICE_COUNT_TOPICS => OPT.topcount
  IF OPT.ha <= 10 THEN
    20.0 => OPT.grundgebuehr
    5.0 OPT.topcount * => OPT.ebenengebuehr
  ELSIF OPT.ha <= 300 THEN
    100.0 => OPT.grundgebuehr
    25.0 OPT.topcount * => OPT.ebenengebuehr
  ELSE
    200.0 => OPT.grundgebuehr
    50.0 OPT.topcount * => OPT.ebenengebuehr
  END_IF
  OPT.grundgebuehr OPT.ebenengebuehr + => OPT.price
END_PROCEDURE

! Hauptprogramm

'price.log' PRICE_SET_LOG ! Logdatei für Fehlermeldungen setzen
PRICE_DISPLAY_INPUT ! Alle Inputparameter anzeigen
PRICE_CREATE_OUTPUT ! Leeres price.txt erzeugen

! calculate price
CALCULATE_PRICE
! header
OPT.price TO_STRING PRICE_WRITELN ! 1. Zeile price.txt mit Preis
'SFr' PRICE_WRITELN ! 2. Zeile price.txt mit Währung

! info
OPT.ha TO_STRING . ' ha' 14 LJUST . ': '
  OPT.grundgebuehr FORMAT_PRICE APP PRICE_WRITELN ! 1. Ausgabe

OPT.topcount TO_STRING . ' Ebenen' 14 LJUST . ': '
  OPT.ebenengebuehr FORMAT_PRICE APP PRICE_WRITELN ! 2. Ausgabe

'-----' PRICE_WRITELN ! etc.
'Zwischentotal' 14 LJUST . ': '
OPT.price FORMAT_PRICE APP PRICE_WRITELN
'-----' PRICE_WRITELN

'Demorabatt' 14 LJUST . ': '
  OPT.price -1.0 * FORMAT_PRICE APP PRICE_WRITELN

'Netto' 14 LJUST . ': '
  0.0 FORMAT_PRICE APP PRICE_WRITELN

'MwSt. 7.6%' 14 LJUST . ': '
  0.0 FORMAT_PRICE APP PRICE_WRITELN

'-----' PRICE_WRITELN

'Total' 14 LJUST . ': '
  0.0 FORMAT_PRICE APP PRICE_WRITELN

'===== ' PRICE_WRITELN

PRICE_CLOSE_OUTPUT ! price.txt schliessen

```

Ein komplexeres Beispiel (nach Modell Buschor) ist im Skript `\user\script\price\zh.cfg` enthalten (inkl. Schnitt mit Tarifzonen).

## 5.5. Angabe des Preisskript in der Produktdatei

Schlussendlich muss der Preisskript (hier z.B. `\script\price\price.cfg`) einem GeoShop Datenprodukt zugeordnet werden (hier z.B. `\user\products\dxfg.pro`). Dazu muss man in der `.pro` Produktdatei folgenden zusätzlichen Eintrag machen (fett markiert):

```
PRODUCT
  name STRING dxf
  display_name STRING GEOBAU2/DXF
  models LIST
    MODEL
      name STRING Grunddatensatz
      display_name STRING 'amtl. Vermessung'
      topics LIST
        STRING Fixpunkte
        STRING Liegenschaften
      }
    }
  MODEL
    name STRING SIA405_mit_Erweiterungen
    display_name STRING 'SIA Werkleitungen'
    topics LIST
      STRING SIA405_Abwasser
      STRING SIA405_Wasser
      STRING SIA405_Kabelkommunikation
    }
  }
  params MAP
  }
  services MAP
    Grunddatensatz STRING il2dxf
    SIA405_mit_Erweiterungen STRING il2dxf
  }
  price_function STRING \script\price\price.cfg
}
```

Die vollständige Syntax für den `price_function` Eintrag lautet:

```
price_function STRING <script>,<layer>
```

Falls der Preisskript auf die Berechnung eines Schnitts von Bestellpolygon mit den Tarifzonen bzw. einer Punktebene angewiesen ist, muss in `<layer>` der entsprechende Layer angegeben werden, in welchem die `PolygonRate` bzw. `PointRate` Objekte enthalten sind.



Nur wenn dem ausgewählten Produkt ein Preisskript zugeordnet wurde, wird dem Benutzer das Menü `Bestellen>Preis berechnen` angezeigt.

# A. Literaturverzeichnis

[1] infoGrips GmbH, GeoShop Benutzerhandbuch

[2] infoGrips GmbH, GEOEDIT Benutzerhandbuch

[3] infoGrips GmbH, iG/Script Benutzer und Referenzhandbuch